

Simulating Quantified Class Constraints

Valery Trifonov
Department of Computer Science
Yale University
New Haven, CT, USA 06520-8285
trifonov@cs.yale.edu

ABSTRACT

Defining nontrivial class instances for irregular and exponential datatypes in Haskell is challenging, and as a solution it has been proposed to extend the language with quantified class constraints of the form $\forall a. C a \Rightarrow C' (f a)$ in the contexts of instance declarations. We show how to express the equivalent of such constraints in vanilla Haskell 98, but their utility in this language is limited. We also present a more flexible solution, which relies on a widely-supported language extension.

Categories and Subject Descriptors

D.3.3 [Programming Languages]: Language Constructs and Features—*Haskell, type classes*

General Terms

Languages, Design

Keywords

Dictionaries, polymorphic types, type classes

1. INTRODUCTION

Contexts in Haskell instance declarations constrain type variables appearing in the defined instance type. As an example (adapted from Ralf Hinze and Simon Peyton Jones [4]) consider the class of types with representation in binary:

```
data Bit = Zero | One
class Binary a where
  showBin :: a → [Bit]
instance Binary Bit where
  showBin = (: [])
```

An instance of *Binary* for lists could be defined as follows:

```
instance Binary a ⇒ Binary [a] where
  showBin = concat . map showBin
```

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. To copy otherwise, to republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee.

Haskell'03, August 25, 2003, Uppsala, Sweden.
Copyright 2003 ACM 1-58113-758-3/03/0008 ...\$5.00.

This instance declaration actually represents an “instance generator,” or a proof that the type $[a]$ is an instance of *Binary* whenever a is; hence the type variable a can be thought of as universally quantified, and the instance declaration as a proof of $\forall a. \text{Binary } a \Rightarrow \text{Binary } [a]$, where the quantification over a is made explicit.

Explicit quantification is not allowed in Haskell contexts, and typically it is unnecessary, because usually the construction of only finitely many class instances must be ensured in order to type-check a Haskell expression. In these cases, as in the above example, the range of the quantifier would include both the class context of the declaration and the type being declared as an instance.

However the ability to express polymorphic recursion in Haskell (directly or in the guise of recursive instance declarations) introduces some cases when specifying class contexts with local quantifiers appears to be the only solution. One of these cases occurs when in order to type-check an instance declaration, we have to prove the existence of infinitely many instances, as in instances of the following types (an example due to David Feuer and Simon Peyton Jones [6], arising in the context of Chris Okasaki’s use of irregular datatypes to represent square matrices in [8]):

```
newtype Two f a = Two (f (f a))
data Sq f a = M a (f a) | E (Sq (Two f) a)
```

In order to define the binary representation for expressions of type *Sq f a* in terms of their components, we need types a and $f a$ to be instances of *Binary*:

```
instance (Binary a, Binary (f a), ...) ⇒ Binary (Sq f a)
where
  showBin (M x xs) = showBin x ++ showBin xs
  showBin (E p)    = showBin p
```

Let us ignore the fact that the constraint on $f a$ is not in Haskell 98 (allowing type expressions in class constraints hinders instance inference); several implementations support such constraints. This instance declaration is not yet complete—additional context, denoted by ellipsis, is needed for the case when the term is constructed with E : we need an instance of *Binary* for $Sq (Two f) a$, which is a substitution instance of the very type for which we are defining the current instance. The instance inference algorithm assumes the current declaration is already available, so we can instantiate the known thus far context of the declaration to find out that we need the instance *Binary (Two f a)*:

```
instance Binary (f (f a)) ⇒ Binary (Two f a) where
  showBin (Two x) = showBin x
```

Now we run into the real problem in defining the instance $\text{Binary}(\text{Sq } f a)$: Since the constraint $\text{Binary}(f(f a))$ must be included in the ellipsis, the instantiation with $\text{Two } f$ for f in turn requires the instance $\text{Binary}(\text{Sq}(\text{Two}(\text{Two } f)) a)$, hence the constraint $\text{Binary}(f(f(f(f a))))$ must be added, etc., ad infinitum¹—meaning that no finite proof exists that an instance $\text{Binary}(\text{Sq } f a)$ can be constructed.

In other cases a finite number of instances would suffice for checking the instance declaration, but no finite instance construction is possible. This happens in the following example (adapted from Peyton Jones’ message [5] and his paper with Hinze [4]):

```
data GRose f a = GBranch a (f (GRose f a))
instance (Binary a, Binary (f (GRose f a))) -- illegal
  => Binary (GRose f a) where
  showBin (GBranch x xs) = showBin x ++ showBin xs
```

The constraint $\text{Binary}(f(\text{GRose } f a))$, required by the second application of showBin , is not legal in Haskell 98. Implementations which allow it (in extensions) accept this declaration; however, except in degenerate cases for f , no instances $\text{Binary}(\text{GRose } f a)$ can actually be created. Consider the example of $f = []$. To create an instance of Binary for $\text{GRose } [] a$, the compiler must first create one for $[\text{GRose } [] a]$; however according to the instance declaration for $\text{Binary } [a]$ it must first create an instance for $\text{GRose } [] a$, causing the instance generation to diverge.

The problems with generating an unbounded number of instances and with mutually dependent instances could be resolved if, instead of trying to describe them all, we could describe a *recipe* for creating them. Hinze and Peyton Jones observe this in [4], and point out that “no ordinary Haskell context will do” and that a solution would be to allow “polymorphic predicates” of the form

$$\text{ctx} ::= \forall \bar{a}. (\text{ctx}_1, \dots, \text{ctx}_n) \Rightarrow C t$$

where C is a class name and t is a type, in instance contexts. (We use the term “quantified constraints” instead of “polymorphic predicates.”) In the above examples the necessary constraint is

$$\forall a. \text{Binary } a \Rightarrow \text{Binary} (f a)$$

Thus the instance declaration for GRose takes the form

```
instance (Binary a, \forall b. Binary b => Binary (f b))
  => Binary (GRose f a) where
  showBin (GBranch x xs) = showBin x ++ showBin xs
```

Now the required instance for $f(\text{GRose } f a)$ can be constructed by instantiating the quantified constraint with the type $\text{GRose } f a$ and applying the result to the current instance. Similarly the instance $\text{Binary}(f a)$, needed in the declaration of $\text{Binary}(\text{Sq } f a)$ in the earlier example, can be constructed from $\text{Binary } a$ (instead of required in the context), thus cutting the infinite chain of required instances.

In this paper we show that Haskell’s constructor classes offer a way to express the equivalent of quantified constraints in vanilla Haskell 98. The full compliance with the language

¹In contrast, other uses of polymorphic recursion require a statically unbounded number of instances to be constructed at run time, but only a finite number of class constraints, so they are correct programs in Haskell 98.

comes at the price of non-local flow-based program transformations, which limit its scope of applicability. We also show how to achieve closer simulation of the uses of quantified constraints when programming with the widely-supported language extension with variable constructor heads in declared instance types. While not covering the range of applications targeted by the proposal for direct language support, these solutions can be applied successfully to problems which have received attention in the community [4, 5, 6, 7, 10], and yield programs in supported Haskell.

2. DICTIONARIES AND TYPE EQUIVALENCES

A quantified constraint $\forall a. C a \Rightarrow C (f a)$ indicates the requirement that an “instance generator” for class C is available for type f . Our goal is thus to allow the context of one instance generator (e.g. for the instance $\text{Binary}(\text{GRose } f a)$) to request the existence of another instance generator (e.g. that for $\forall b. \text{Binary } b \Rightarrow \text{Binary} (f b)$), whereas Haskell only allows instances to be requested. With this observation, let us look at the semantics of instances and generators and try to find a correspondence.

The standard semantics of Haskell type classes is given by translating the language into one with explicit passing of *dictionaries* [9] — records containing the methods of the required instances. Thus a class declaration

```
class C a where  $\overline{m_i} :: \text{cty}_i$ 
```

gives rise to a type constructor $C = \lambda a :: \kappa. \{\overline{m_i} :: \llbracket \text{cty}_i \rrbracket_a\}$ in a standard extension of F_ω [2] with records (tuples):

```
kinds   $\kappa ::= * \mid \kappa \rightarrow \kappa'$ 
types   $\tau ::= a \mid \lambda a :: \kappa. \tau \mid \tau \tau' \mid \tau \rightarrow \tau' \mid \forall a :: \kappa. \tau$ 
         $\mid \{\overline{m_i} :: \overline{\tau_i}\} \mid \dots$ 
terms   $e ::= x \mid \lambda x :: \tau. e \mid e e' \mid \Lambda a :: \kappa. e \mid e[\tau]$ 
         $\mid \{\overline{m_i} = \overline{e_i}\} \mid e.m \mid \dots$ 
```

where x ranges over term variables, a and b range over type variables, and m ranges over labels.² The types $\llbracket \text{cty}_i \rrbracket_a$ are the translations of cty_i , which are Haskell types with contexts; the most important feature of the translation is that it turns class contexts into types of dictionaries as arguments, and quantifies over all free type variables other than a :

$$\llbracket (C_j(a_j \overline{t_j})) \Rightarrow t \rrbracket_a = \forall \overline{b}. \overline{C_j(a_j \overline{t_j})} \rightarrow t$$

where \overline{b} is a sequence of all type variables in the set $\{\overline{a_j}\} \cup FV(\overline{t_j}) \cup FV(t) - \{a\}$, i.e. all type variables free in the type (including the contexts) but a . Note that the metavariable t ranges over simple Haskell types; we gloss over the details of their translation by assuming they are a subset of the target language types τ .

An “instance generator” declaration, of the form

```
instance  $(\overline{C_j a_j}) \Rightarrow C \tau$  where  $\overline{m_i} = \overline{e_i}$ 
```

can be translated as a “dictionary generator” term

$$\underline{dg} = \Lambda \overline{b}. \lambda \overline{d_j} :: \overline{C_j a_j}. \{\overline{m_i} = \llbracket \overline{e_i} \rrbracket_{\overline{C_j a_j}}^{d_j}\}$$

²We use an overloaded notation \overline{A} for sequences of terms of the syntactic category ranged over by A : the separators between the terms in the sequence should be inferred from the context. If each of the terms A has component subterms we need to refer to, they are all indexed with the same subscript; in some cases these subterms are themselves sequences.

where the translation $[\cdot]_{C_j a_j}^{d_j}$ replaces all uses of the instances $C_j a_j$ by operations on the corresponding dictionaries d_j . Details of this translation are omitted, because they are not important for us at this point; important is the type of the term dg :

$$dg :: \forall \bar{b}. \overline{C_j a_j} \rightarrow \{\overline{m_i} :: \overline{\tau_i}\}$$

Thus, in terms of this translation, our goal is to make dictionary generators like dg take parameters of the type of dg . However the translation obviously only allows dictionaries as parameters of dictionary generators.

But perhaps it is possible to have a dictionary parameter whose type is *isomorphic* to the type of a generator?

The difference between dictionaries and dictionary generators is that the former are records of values, while the latter are polymorphic functions producing records of values. However there are well-known isomorphisms which we can use to construct maps between the two types, namely the distributivity laws

$$\begin{aligned} \tau &\rightarrow \overline{\{m_i :: \tau_i\}} &\leftrightarrow &\overline{\{m_i :: \tau \rightarrow \tau_i\}} \\ \forall a :: \kappa. \overline{\{m_i :: \tau_i\}} &\leftrightarrow &\overline{\{m_i :: \forall a :: \kappa. \tau_i\}} \end{aligned}$$

So we have

$$\forall \bar{a}. \overline{C_j a_j} \rightarrow \{\overline{m_i} :: \overline{\tau_i}\} \leftrightarrow \overline{\{m_i :: \forall \bar{a}. \overline{C_j a_j} \rightarrow \tau_i\}}$$

This is a result in our variant of F_ω , but not in Haskell yet—not all F_ω types can be represented in Haskell. In particular, the argument types we must push under the record type constructor correspond to dictionaries, and hence to contexts in Haskell—that is, they cannot be represented as parameters of Haskell functions. Luckily, however, methods in Haskell 98 can have *local contexts* in addition to the context of the instance declaration, and the prenex universal quantification on method types corresponds exactly to the quantification in the type on the right hand side.

3. A REPRESENTATION IN HASKELL 98

Returning to Haskell, suppose we have a class declaration

$$\text{class } C \text{ } a \text{ where } \overline{m_i} :: \overline{ctx_i} \Rightarrow t_i,$$

and in the context of some instance declaration we need the quantified constraint $\forall b. \overline{ctx'} \Rightarrow C(f t)$, where the type variable b appears in $\overline{ctx'}$ and the type t :

$$\text{instance } (\forall b. \overline{ctx'} \Rightarrow C(f t), \overline{ctx''} \Rightarrow C' t' \text{ where } \overline{m'_j} = e_j$$

We introduce a “functorial class” C_f declared as

$$\text{class } C_f \text{ } f \text{ where } \overline{m_f i} :: \forall b. [f t/a](\overline{ctx'}, ctx_i) \Rightarrow t_i$$

where $[t/a]cty$ denotes the type obtained by substituting t for a in cty ; the quantification over b is implicit in Haskell 98 but shown here for emphasis, while other implicitly quantified variables are not shown. Then we use the constraint $C_f f$ instead of the desired quantified constraint, and we use the method names $m_f i$ instead of m_i in the expressions in the dynamic scope of this constraint, e.g.

$$\text{instance } (C_f f, \overline{ctx''}) \Rightarrow C' t' \text{ where } \overline{m'_j} = [m_f i/m_i]e_j$$

This syntactic transformation is in general non-local: The requirement to cover the dynamic scope of the constraint implies that all overloaded functions with the constraint $C a$ in

the type, instantiated with $f t$ for a in applications reachable from e_j , must be cloned, and the names of these functions and m_i substituted by their clones’ names in the cloned code.

We also provide instances of the form

$$\text{instance } C_f \text{ } T \text{ where } \overline{m_f i} = \overline{m_i}$$

for each type constructor T for which we would need an instance of the desired quantified constraint. The methods in these instances are (modulo the type isomorphisms of Section 2) essentially trampolines to the defined as usual methods in instances of C for applications of T .

Although we omit the kind specifications of type variables for brevity, it should be clear that this transformation is valid for arbitrary consistent kinds; however different “functorial classes” must be provided for type constructors of different kinds. Since this scheme supports the cases when the type t in the quantified constraint $\forall b. \overline{ctx'} \Rightarrow C(f t)$ is not simply the variable b , if the constraints $\forall b. ctx_1 \Rightarrow C(f_1 t_1)$ and $\forall b. ctx_2 \Rightarrow C(f_2 t_2)$ are both needed in instance declarations, and $t_1 \neq t_2$, we would have different “functorial classes” for them; this is also the case in particular when the kinds of t_1 and t_2 (hence of f_1 and f_2) are different.

In the example of the *GRose* type in the introduction, Hinze and Peyton Jones suggest the use of the quantified constraint $\forall a. \text{Binary } a \Rightarrow \text{Binary } (f a)$ to define an instance of *Binary*. We instead declare the class

$$\begin{aligned} \text{class } \text{Binary_}f \text{ } f \text{ where} \\ \text{showBin_}f :: \text{Binary } a \Rightarrow f a \rightarrow [\text{Bit}] \end{aligned}$$

Then an instance of *Binary* can be constructed for *GRose* as follows:

$$\begin{aligned} \text{instance } (\text{Binary } a, \text{Binary_}f \text{ } f) \\ \Rightarrow \text{Binary } (\text{GRose } f a) \text{ where} \\ \text{showBin } (\text{GBranch } x \text{ } xs) = \\ \text{showBin } x \text{ } ++ \text{showBin_}f \text{ } xs \end{aligned} \quad (1)$$

Additionally, for the construction of *Binary* instances for *GRose* $[\] \text{Bit}$ we need also the declaration

$$\begin{aligned} \text{instance } \text{Binary_}f \text{ } [\] \text{ where} \\ \text{showBin_}f = \text{showBin} \end{aligned}$$

assuming we already have the instance *Binary* $[a]$, shown in the introduction.

The simplicity of the auxiliary declarations is due to the type inference and dictionary conversion, which automatically insert the type and dictionary applications. As an illustration, the translation of the above code into a variant of F_ω is shown in Figure 1; the calculus is enriched with pattern matching on function arguments and a fixpoint expression $\text{rec } x :: \tau = e$ to allow the translation of recursive instances, and we assume the standard definitions of *List*, *concat*, *map*, *append*, and *Bit* are available. Note that the definition of *Binary_}f List* implements half of the isomorphism between *Binary_}f f* and $\forall a. \text{Binary } a \rightarrow \text{Binary } (f a)$, while the other half is inlined in the last three lines of the figure and evident in the order of the selection from and applications of d_f . (An implementation based on Hinze and Peyton Jones’ proposal would just avoid these shuffles.)

This approach, however, is limited by the non-local aspects of the transformation. To apply it, we must be able to locate and clone statically all functions which are invoked from the translated instance declaration and have types with the constraints we are replacing. Since Haskell 98 does not

```

Binary :: * → *
  = λa :: *. {showBin :: a → List Bit}

Binary_f :: (* → *) → *
  = λf :: * → *.
    {showBin_f :: ∀a. Binary a → f a → List Bit}

Binary_List :: ∀a :: *. Binary a → Binary (List a)
  = Λa :: *. λd_a :: Binary a.
    {showBin = λxs :: List a.
      concat [Bit]
        (map [a] [List Bit] (d_a.showBin) xs)}

Binary_f_List :: Binary_f List
  = {showBin_f = Λa :: *. λd_a :: Binary a.
    (Binary_List [a] d_a).showBin}

Binary_GRose_type :: *
  = ∀a. ∀f. Binary a → Binary_f f → Binary (f a)

Binary_GRose :: Binary_GRose_type
  = rec d :: Binary_GRose_type
    = Λa. Λf. λd_a. λd_f.
      {showBin =
        λ(GBranch (x :: a) (xs :: f (GRose f a))).
          append [Bit]
            (d_a.showBin x)
            (d_f.showBin_f [GRose f a]
              (d [a] [f] d_a d_f)
              xs)}

```

Figure 1: Translation of instances of *Binary*.

allow constraints to be nested in types, it may appear that these functions are not first class, hence their invocations are always direct and their reachability can be determined statically. This is not the case, because these functions may be methods of another class; then their types may contain constraints,³ and their invocations are not only indirect—they are invisible in the Haskell code. Consider the types *Sq* and *Two*, introduced earlier. Defining an instance of *Binary* for *Sq* is now straightforward by replacing the quantified constraint on *f* with *Binary_f f*. We must then define an instance of *Binary_f* for *Two f* under the assumption of *Binary_f f*. However, this is impossible, because (following the algorithm) we need to replace with *Binary_f f* the constraint *Binary a* in the type of *showBin_f* in the assumed instance *Binary_f f*, which cannot be determined statically.

4. A MORE FLEXIBLE APPROACH

Suppose we also need an instance of *Binary* for the type *GRose (GRose []) Bit*. To satisfy the constraints in declaration (1), we have to declare an instance of *Binary_f* for *GRose []*. Naturally we can obtain it from the more general

```

instance Binary_f f ⇒ Binary_f (GRose f) where
  showBin_f = showBin

```

Just as in the case of lists above, this definition exploits the existence of an instance of *Binary* for *GRose f a*. However we have to provide these (trivial) declarations, each defining *showBin_f* in terms of *showBin*, for each type construc-

³Ironically this is exactly the Haskell feature that made possible the approach in the first place.

tor required to satisfy the quantified constraint encoded by *Binary_f*, and as we showed they introduce a major weakness, because their invocations cannot be replaced statically.

An alternative is to define *showBin* in terms of *showBin_f* (to illustrate this we have to ignore the code shown above, including and following (1), as well as the earlier instance declaration for *Binary [a]*). It turns out that a single declaration suffices:

```

instance (Binary a, Binary_f f) ⇒ Binary (f a) where
  showBin = showBin_f

```

Unfortunately, due to the type variable *f* in the head of the instance type, this declaration is not in Haskell 98; however, at least two implementations support extensions allowing such declarations. The list type constructor is now handled by one additional declaration:

```

instance Binary_f [] where
  showBin_f = concat . map showBin

```

An analogous declaration would do it for *GRose*, but its kind suggests that a more general definition is useful:

```

class Binary_f3 (g :: (* → *) → * → *) where
  showBin_f3 :: (Binary a, Binary_f f) ⇒ g f a → [Bit]

instance (Binary_f (f :: * → *), Binary_f3 g)
  ⇒ Binary_f (g f) where
  showBin_f = showBin_f3

```

```

instance Binary_f3 GRose where
  showBin_f3 (GBranch x xs) = showBin x ++ showBin xs

```

The kind annotations are shown for clarity, but they are inferred unambiguously. The strong similarity between the instance declarations for *Binary_f (g f)* and *Binary (f a)*, as well as those for other function kinds, cannot be taken advantage of in Haskell, because they refer to classes with different (names and) types of methods.⁴ On the other hand we only have to define one class and one instance for every kind of type constructor for which we need instances of *Binary*, and its subkinds (i.e. syntactic subterms of the kind expression), and in a typical Haskell program their number is very small.

With this approach the type *Sq*, shown in the introduction, is just as easy to handle:

```

instance Binary_f3 Two where
  showBin_f3 (Two x) = showBin x

instance Binary_f3 Sq where
  showBin_f3 (M x xs) = showBin x ++ showBin xs
  showBin_f3 (E s)   = showBin s

```

In another example, that of an “exponential” type

```

data T f a = A a (f a) | T (T f (T f a))

```

the encoding works together with Haskell’s recursive instances:

```

instance Binary_f3 T where
  showBin_f3 (A x xs) = showBin x ++ showBin xs
  showBin_f3 (T u)   = showBin u

```

The syntax of quantified constraints allows for an empty list of premises, as in for instance $\forall a. C (f a)$. A case when

⁴An extension of Clean which allows sharing the code for such instances is presented in [1]; it can be supported by compiling to the language of [3].

this sort of constraint is useful was demonstrated by Ashley Yakeley in [10]: The class of bifunctors

```
class Bifunctor f where
  bimap :: (a → a') → (b → b') → f a' b → f a b'
```

can be synthesized from the classes of functors and cofunctors:

```
class Functor f where -- standard
  fmap :: (a → b) → f a → f b
class Cofunctor2 f where
  comap2 :: (a → a') → f a' b → f a b
```

if one could write the instance declaration

```
instance (Cofunctor2 f, ∀a. Functor (f a))
  ⇒ Bifunctor f where
  bimap fa fb = comap2 fa . fmap fb
```

Following the approach, we write instead

```
class Functor_f f where
  fmap_f :: (a → a') → f b a → f b a'
instance Functor_f f ⇒ Functor (f a) where
  fmap = fmap_f
instance (Cofunctor2 f, Functor_f f)
  ⇒ Bifunctor f where
  bimap fa fb = comap2 fa . fmap_f fb
```

which completes a program valid in Haskell with extensions for variable head instances and overlapping instances.

5. RELATED WORK

Ralf Hinze and Simon Peyton Jones describe in [4] the utility of quantified constraints in the context of automatic derivation of instances. They propose extending the language with quantified constraints, and provide semantics for the extension. Artem Alimarine and Rinus Plasmeijer [1] present extensions to Clean, which allow the use of induction on the structure of kinds in the definition of classes and instances in the style of [3].

In contrast the simulations outlined in our paper are not intended as a substitute for a language extension for the purpose of providing compiler support for other features (for example automatic instance derivation), although our second approach can be used as a basis for a preprocessor. Our goal is to offer a solution for problems involving a limited set of kinds, for which it is feasible to code the required class and instance declarations. Such problems, requiring quantified constraints, have been discussed multiple times on the Haskell mailing lists in recent years. Conor McBride [7] has independently outlined the essence of the solution presented here; unfortunately subsequent discussions on the same topic indicate that his description was not interpreted to suggest a solution within the existing language.

6. CONCLUSION

Of the two presented approaches to simulating quantified constraints, the first has the advantages that it can be used in Haskell 98, and it does not require changes in the way instances of the original classes are constructed. However its dependence on the ability to perform (a restricted form of) flow analysis of the program prevents it from handling some cases of irregular types. The second approach requires an

extension of Haskell allowing a type variable in the head of an instance declaration, and forces some changes in the style of coding of instances; in return it is much more flexible. While not a substitute for a language extension, the second approach appears quite useful in solving typical problems involving quantified constraints.

7. ACKNOWLEDGMENTS

Thanks to Paul Hudak and the anonymous referees for their suggestions on improving the presentation and the technical content.

This research was supported in part by DARPA OASIS grant F30602-99-1-0519, NSF grant CCR-0208618, and NSF ITR grant CCR-0081590. Any opinions, findings, and conclusions contained in this document are those of the author and do not reflect the views of these agencies.

8. REFERENCES

- [1] Artem Alimarine and Rinus Plasmeijer. A generic programming extension for Clean. In *Implementation of Functional Languages, 13th International Workshop (IFL 2001)*, volume 2312 of *LNCS*, pages 168–186, Stockholm, Sweden, September 2001. Springer.
- [2] J. Y. Girard. *Interpretation Fonctionnelle et Elimination des Coupures dans l'Arithmétique d'Ordre Supérieur*. PhD thesis, University of Paris VII, 1972.
- [3] Ralf Hinze. Polytypic values possess polykinded types. *Science of Computer Programming*, 43:129–159, 2002.
- [4] Ralf Hinze and Simon Peyton Jones. Derivable type classes. In *Proceedings of the 2000 ACM SIGPLAN Haskell Workshop*, volume 41 of *Electronic Notes in Theoretical Computer Science*, Montréal, Canada, September 2000. Elsevier. <http://www.elsevier.nl/inca/publications/store/5/0/5/6/2/5/>.
- [5] Simon Peyton Jones. Re: Rank-2 polymorphism & type inference, December 2000. Message posted on the Haskell mailing list, available at <http://haskell.cs.yale.edu/pipermail/haskell/2000-December/006303.html>.
- [6] Simon Peyton Jones. Deriving excitement, February 2002. Message posted on the Haskell mailing list, available at <http://haskell.cs.yale.edu/pipermail/haskell/2002-February/005426.html>.
- [7] Conor T. McBride. Re: higher-kind deriving... or not, February 2002. Message posted on the Haskell-café mailing list, available at <http://www.haskell.org/pipermail/haskell-cafe/2002-February/002785.html>.
- [8] Chris Okasaki. From fast exponentiation to square matrices: An adventure in types. In *Proceedings of the 1999 ACM SIGPLAN International Conference on Functional Programming*, pages 28–35, Paris, France, September 1999.
- [9] John Peterson and Mark P. Jones. Implementing type classes. In *Proceedings of the 1993 ACM SIGPLAN Symposium on Programming Language Design and Implementation*, pages 227–236, Albuquerque, New Mexico, June 1993.
- [10] Ashley Yakeley. Re: Arrow classes, July 2003. Message posted on the Haskell-café mailing list, available at <http://haskell.cs.yale.edu/pipermail/haskell-cafe/2003-July/004659.html>.